

计算器的设计

3.1 功能要求

基于 80C51 单片机的计算器控制系统的设计功能要求如下。

(1) 计算器能正常显示输入数据和计算结果。

(2) 计算器能对 2 位十进制整数进行简单的加、减、乘、除四则运算,在做除法运算时能自动舍去商的小数部分。

3.2 系统设计

基于单片机的计算器控制系统以 80C51 单片机为控制中心,由电源电路、按键电路、时钟电路、数码显示电路和复位电路等几部分组成。系统框图如图 3-1 所示。




图 3-1 基于 80C51 单片机的计算器控制系统框图

3.3 相关知识

在本案例中需要通过学习和查阅资料,了解和掌握以下方面的知识。

- (1) 80C51 单片机的工作原理。
- (2) 复位电路工作原理及电路设计。
- (3) 时钟电路工作原理及电路设计。
- (4) 按键电路的设计。
- (5) 七段数码显示的原理及使用。
- (6) 计算模块程序设计。




图 3-2 基于 80C51 单片机的计算器控制系统电路原理图

3.4 硬件设计

3.4.1 电路原理图

基于 80C51 单片机的计算器控制系统电路原理图如图 3-2 所示。

1. 数据显示模块

4 位共阳极的七段数码管用于显示当前数值的千、百、十、个位,采用动态显示方式,P1 口输出数码管的字形码,P0 口输出数码管的字位码。

2. 按键模块

键盘由 16 个按键组成,包括 10 个数字键(由 0~9 组成)、5 个运算符号键(加、减、乘、除、等于)和 1 个清除键(作用相当于整体复位)。80C51 单片机的 P2 口作为键盘口,其中 P2.4~P2.7 口为键盘扫描输出线,P2.0~P2.3 口为键盘扫描输入线。

3.4.2 元器件清单

基于 80C51 单片机的计算器控制系统元器件清单见表 3-1。

表 3-1 基于 80C51 单片机的计算器控制系统元器件清单

元器件名称	型 号	数量	用 途	元器件名称	型 号	数量	用 途
单片机	80C51	1	控制器	按键	SW-PB	1	复位按键
晶体振荡器	12MHz	1	时钟电路	排阻	8.8k Ω ×4	1	上拉电阻
电容	30pF	1	时钟电路	LED 数码管	四联共阴	1	数码显示
电解电容	22 μ F/10V	1	复位电路	按键	KEYPAD-SMALLCALC	1	按键电路
电阻	1k Ω	1					

3.5 软件设计

3.5.1 程序流程图

基于 80C51 单片机的计算器控制系统主要分为显示模块、按键扫描模块、计算模块。其程序执行流程图如图 3-3 所示。

3.5.2 程序清单

基于 80C51 单片机的计算机控制器程序清单如下:

```

; 基于 80C51 单片机的计算器控制存储单元分配
; 30H: 数值个位显示单元; 31H: 数值十位显示单元; 32H: 数值百位显示单元; 33H: 数值千位
; 显示单元; 23H: 第一操作数存储单元; 24H: 第二操作数存储单元; 25H: 键值暂存单元;
; 27H: 清除键状态; 34H~37H: 结果数据转换暂存单元; 38H,39H: 结果高、低 8 位暂存单元;
; R5: 操作数计数单元; R4: 操作数数值位数计数单元; R3: 运算符号存储单元
      DISPB  EQU      30H          ; 显示缓冲区首地址定义
      ORG    0000H                ; 主程序的入口地址
START:  MOV     DISPB,  #0C0H      ; 4 个数码管显示 0000 字形数据

```




图 3-3 程序流程图

```

MOV    DISPB+1, #0C0H
MOV    DISPB+2, #0C0H
MOV    DISPB+3, #0C0H
MOV    37H,     #00H           ; 数值初始化
MOV    27H,     #00H
MOV    26H,     #00H
MOV    25H,     #00H
MOV    24H,     #00H
MOV    23H,     #00H
MOV    R3,      #00H
MOV    R5,      #00H
MOV    R4,      #00H
LOOP:  LCALL    KEY             ; 调用键盘扫描子程序
MOV    R6,      27H
CJNE   R6,      #00H, START    ; 清除键判断
MOV    R6,      26H
CJNE   R6,      #00H, FA       ; 显示数据转换子程序
LCALL   CONV     ; 调用数码管显示数据转换子程序
FA:    LCALL   DISPS           ; 调用数码管动态显示子程序
        SJMP   LOOP
KEY:    MOV     P2,      #0FH   ; 键盘扫描子程序
        MOV     A,      P2

```

```

        ANL    A,      #0FH
        CJNE   A,      #0FH, K10
        RET
K10:    JB     P2.0,    K20
K11:    MOV    P2,      #0EFH
        JB     P2.0,    K12
        CJNE   R5,      #00H, L1
        MOV    R4,      #00H
        INC    R5
        MOV    R3,      #1H
L1:     LJMP   CEND
K12:    MOV    P2,      #0DFH
        JB     P2.0,    K13
        CJNE   R5,      #00H, L2
        MOV    R4,      #00H
        INC    R5
        MOV    R3,      #2H
L2:     LJMP   CEND
K13:    MOV    P2,      #0BFH
        JB     P2.0,    K14
        CJNE   R5,      #00H, L3
        MOV    R4,      #00H
        INC    R5
        MOV    R3,      #3H
L3:     LJMP   CEND
K14:    MOV    P2,      #7FH
        JB     P2.0,    K1L
        CJNE   R5,      #00H, L4
        MOV    R4,      #00H
        INC    R5
        MOV    R3,      #4H
K1L:    LJMP   KEND
L4:     LJMP   CEND
K20:    JB     P2.1,    K30
K21:    MOV    P2,      #0EFH
        JB     P2.1,    K22
        LJMP   DYU
K22:    MOV    P2,      #0DFH
        JB     P2.1,    K23
        MOV    25H,     #3H
        LJMP   OPR
K23:    MOV    P2,      #0BFH
        JB     P2.1,    K24
        MOV    25H,     #6H
        LJMP   OPR
K24:    MOV    P2,      #7FH
        JB     P2.1,    CC
        MOV    25H,     #9H
        LJMP   OPR

```

```

CC:      LJMP    KEND
K30:     JB      P2.2,    K40
K31:     MOV     P2,      #0EFH
         JB      P2.2,    K32
         MOV     25H,     #0H
         LJMP    OPR
K32:     MOV     P2,      #0DFH
         JB      P2.2,    K33
         MOV     25H,     #2H
         LJMP    OPR
K33:     MOV     P2,      #0BFH
         JB      P2.2,    K34
         MOV     25H,     #5H
         LJMP    OPR
K34:     MOV     P2,      #7FH
         JB      P2.2,    KEND
         MOV     25H,     #8H
         LJMP    OPR
K40:     JB      P2.3,    KEND
K41:     MOV     P2,      #0EFH
         JB      P2.3,    K42
         INC     27H
         LJMP    KEND
K42:     MOV     P2,      #0DFH
         JB      P2.3,    K43
         MOV     25H,     #1H
         LJMP    OPR
K43:     MOV     P2,      #0BFH
         JB      P2.3,    K44
         MOV     25H,     #4H
         LJMP    OPR
K44:     MOV     P2,      #7FH
         JB      P2.3,    KEND
         MOV     25H,     #7H
         LJMP    OPR
OPR:     INC     R4
         CJNE    R5,      #00H, CHANG
         CJNE    R4,      #2H,  KBK
ZHI:     MOV     A,       20H
         MOV     B,       #0AH
         MUL     AB
         MOV     20H,     25H
         ADD     A,       20H
         CJNE    R5,      #00H, CZH
         MOV     23H,     A
         LJMP    KEND
CZH:     MOV     24H,     A
         LJMP    KEND
KEND:    MOV     P2,      #0FH

```

; 键值的存储

```

        MOV     A,      P2
        ANL     A,      #0FH
        CJNE    A,      #0FH, KEND
        RET
CHANG:  CJNE    R4,      #2H, HONG
        AJMP    ZHI
HONG:   JNC     CEND
        INC     R5
        MOV     20H,     25H
        MOV     24H,     25H
        LJMP    CEND
KBK:    JNC     CEND
        MOV     20H,     25H
        MOV     23H,     25H
        LJMP    CEND
DYU:    MOV     R5,      #00H           ; 结果的计算
        MOV     R4,      #00H
        MOV     25H,     #00H
        CJNE    R3,      #1H, S
        MOV     A,      23H
        ADD     A, 2      4H
        MOV     23H,     A
        SJMP    CEND
S:      CJNE    R3,      #2H, S1
        CLR     C
        MOV     A,      23H
        CJNE    A,      #00H, JIANFA
        MOV     23H,     #0EH
        SJMP    CEND
JIANFA: SUBB     A,      4H
        MOV     23H,     A
        SJMP    CEND
S1:     CJNE    R3,      #3H, S2
        MOV     A,      23H
        MOV     B,      24H
        MOV     A,      23H
        MUL     AB
        MOV     R6,      B
        CJNE    R6,      #00H, CHENG
        MOV     23H,     A
        SJMP    CEND
S2:     CJNE    R3,      #4, CEND
        MOV     A,      24H
        CJNE    A,      #00H, ZY
        MOV     A,      #00H
        SJMP    CEND
ZY:     MOV     A,      23H
        MOV     B,      24H
        DIV     AB

```

```

                MOV    23H,    A
CEND:          MOV    P2,      #0FH
                MOV    A,      P2
                ANL    A,      #0FH
                CJNE   A,      #0FH, CEND
                RET
CHENG:         MOV    R3,      B
                MOV    B,      #64H
                DIV    AB
                MOV    36H,    A
                MOV    A,      B
                MOV    B,      #0AH
                DIV    AB
                MOV    35H,    A
                MOV    34H,    B
                MOV    DPTR,   #TABLE
XIAN:          MOV    A,      34H
                ADD    A,      #6H
                MOV    B,      #0AH
                DIV    AB
                MOV    25H,    A
                MOV    34H,    B
                MOV    A,      B
                MOVC   A,      @A+DPTR
                MOV    DISPB,  A
                MOV    A,      35H
                ADD    A,      #5H
                ADD    A,      25H
                MOV    B,      #0AH
                DIV    AB
                MOV    25H,    A
                MOV    35H,    B
                MOV    A,      B
                MOVC   A,      @A+DPTR
                MOV    DISPB+1, A
                MOV    A,      36H
                ADD    A,      #2H
                ADD    A,      25H
                MOV    B,      #0AH
                DIV    AB
                ADD    A,      37H
                MOV    37H,    A
                MOVC   A,      @A+DPTR
                MOV    DISPB+3, A
                MOV    36H,    B
                MOV    A,      B
                MOVC   A,      @A+DPTR
                MOV    DISPB+2, A
                DJNZ   R6,     XIAN

```

```

        MOV     26H,      # 1H
        AJMP    CEND
CONV:   CJNE    R5,      # 02H, BEI      ; 数码管显示数据转换子程序
        MOV     A,      24H
        AJMP    JING
BEI:    MOV     A,      23H
JING:   MOV     B,      # 64H
        DIV     AB
        MOV     DPTR,    # TABLE
        MOVC    A,      @A+DPTR
        MOV     DISPB+2, A
        MOV     A,      B
        MOV     B,      # 0AH
        DIV     AB
        MOVC    A,      @A+DPTR
        MOV     DISPB+1, A
        MOV     A,      B
        MOVC    A,      @A+DPTR
        MOV     DISPB,   A
        RET
DISPS:  MOV     R0,      # DISPB      ; 数码管动态显示子程序
        MOV     R2,      # 4H
        MOV     A,      # 1H
SC:     PUSH    ACC
        MOV     A,      @R0
        MOV     P1,      A
        POP     ACC
        MOV     P0,      A
        LCALL   DELAY1
        RL      A
        INC     R0
        DJNZ    R2,      SC
        MOV     P0,      # 1H
        RET
DEL12:  MOV     R6,      # 1EH
DEL2:   MOV     R7,      # 64H
DEL1:   DJNZ    R7,      DEL1
DJNZ    R6,     DEL2
        RET
DELAY1: MOV     R6,      # 70
D2:     MOV     R7,      # 100
        DJNZ    R7,      $
        DJNZ    R6,      D2
        RET
TABLE:  DB      0C0H,0F9H,0A4H,0B0H,99H    ; 字符表
        DB      92H,82H,0F8H,80H,90H,88H
        DB      83H,0C6H,0A1H,86H,8EH
        END

```




图 3-4 基于单片机的计算器控制系统仿真效果图